

LEAST SQUARES CALCULATIONS WITH GAMS

ERWIN KALVELAGEN

ABSTRACT. This document show how different type of regression models can be solved with GAMS.

1. LINEAR LEAST SQUARES

1.1. **OLS is an optimization problem.** Ordinary Least Squares (OLS) is a technique to estimate parameters in a linear statistical model:

$$(1) \quad y = X\beta + \epsilon$$

where y is the dependent (endogenous) variable (stored as an $(n \times 1)$ vector), and X is an $(n \times k)$ matrix of k independent (exogenous) variables. ϵ is an error term. We assume that $E(\epsilon'\epsilon) = \sigma^2 I_n$, i.e. the different ϵ_i 's are independent. We can estimate β by the optimization model:

$$\begin{aligned} \text{OLS} \quad & \underset{\beta}{\text{minimize}} \quad \sum_i \epsilon_i^2 \\ & \text{subject to } y_i = \sum_j x_{i,j}\beta_j + \epsilon_i \end{aligned}$$

This model is trivially coded in GAMS using a simple linearly constrained NLP. Consider the following data from [25]: we have 40 cross section observations of weekly household expenditure on food and on weekly household income (see table 1). We assume that the ‘consumption function’ is linear. Note that when a constant term is part of the model, a simple approach is to have a column of ones in the X matrix (usually this is the first column).

Notice that the notation is sometimes confusing: in many optimization models, x denotes the primary decision variable. In regression, X is a data matrix (i.e. a parameter in GAMS).

The econometrics package CHAZAM [51] gives the following results using the OLS procedure on this data set:

```
|_SAMPLE 1 40
|_READ (GHJ.DAT) FOOD INCOME

UNIT 88 IS NOW ASSIGNED TO: GHJ.DAT
 2 VARIABLES AND      40 OBSERVATIONS STARTING AT OBS      1

|_OLS FOOD INCOME

OLS ESTIMATION
 40 OBSERVATIONS      DEPENDENT VARIABLE = FOOD
...NOTE..SAMPLE RANGE SET TO:    1,    40

R-SQUARE =     .3171      R-SQUARE ADJUSTED =     .2991
```

Date: December 13, 2007.

food	income	food	income
9.46	25.83	17.77	71.98
10.56	34.31	22.44	72.00
14.81	42.50	22.87	72.23
21.71	46.75	26.52	72.23
22.79	48.29	21.00	73.44
18.19	48.77	37.52	74.25
22.00	49.65	21.69	74.77
18.12	51.94	27.40	76.33
23.13	54.33	30.69	81.02
19.00	54.87	19.56	81.85
19.46	56.46	30.58	82.56
17.83	58.83	41.12	83.33
32.81	59.13	15.38	83.40
22.13	60.73	17.87	91.81
23.46	61.12	25.54	91.81
16.81	63.10	39.00	92.96
21.35	65.96	20.44	95.17
14.87	66.40	30.10	101.40
33.00	70.42	20.90	114.13
25.19	70.48	48.71	115.46

TABLE 1. A household food expenditure data set

```

VARIANCE OF THE ESTIMATE-SIGMA**2 =   46.853
STANDARD ERROR OF THE ESTIMATE-SIGMA =   6.8449
SUM OF SQUARED ERRORS-SSE=   1780.4
MEAN OF DEPENDENT VARIABLE =   23.595
LOG OF THE LIKELIHOOD FUNCTION = -132.672

VARIABLE    ESTIMATED    STANDARD    T-RATIO      PARTIAL STANDARDIZED ELASTICITY
      NAME     COEFFICIENT    ERROR      38 DF    P-VALUE CORR. COEFFICIENT AT MEANS
INCOME      .23225      .5529E-01   4.200     .000   .563   .5631   .6871
CONSTANT    7.3832      4.008      1.842     .073   .286   .0000   .3129
|_STOP

```

Using the simple nonlinear GAMS formulation as displayed in the following fragment:

```

variables
  constant   'estimate constant term coefficient'
  income     'estimate income coefficient'
  residual(i) 'error term'
  sse        'sum of squared errors'
;

equations
  fit(i)     'the linear model'
  obj        'objective function'
;

obj..      sse =e= sum(i, sqr(residual(i)));
fit(i)..  data(i,'expenditure') =e= constant + income * data(i,'income') + residual(i);

model osl1 /obj,fit/;
solve osl1 minimizing sse using nlp;
display constant.l, income.l, sse.l;

```




FIGURE 1. OLS Estimation

we get the following estimates:

VARIABLE constant.L	=	7.383 estimate constant term coefficient
VARIABLE income.L	=	0.232 estimate income coefficient
VARIABLE sse.L	=	1780.413 sum of squared errors

The complete model is reproduced below:

1.1.1. Model ols1.gms.¹

```
$ontext
Ordinary Least Squares (OLS) by minimizing the Sum of
Squared Errors directly.

Erwin Kalvelagen, october 2000

$offtext

set i /i1*i40/;

$include expdata.inc

variables
  constant    'estimate constant term coefficient'
  income      'estimate income coefficient'
  residual(i) 'error term'
  sse         'sum of squared errors'
;

equations
```

¹<http://amsterdamoptimization.com/models/statistics/ols1.gms>

```

      fit(i)      'the linear model'
      obj       'objective'

;

obj..    sse == sum(i, sqr(residual(i)));
fit(i)..  data(i,'expenditure') == constant + income*data(i,'income') + residual(i);

model ols1 /obj,fit/;

solve ols1 minimizing sse using nlp;
display constant.l, income.l, sse.l;

```

The model uses an include for the data:

1.1.2. *Include file expdata.inc.*²

```

* cross-section data: weekly household expenditure on food and
* weekly household income from Griffiths, Hill and Judge,
* 1993, Table 5.2, p. 182.

table data(i, *)
      expenditure income
i1      9.46     25.83
i2     10.56     34.31
i3     14.81     42.50
i4     21.71     46.75
i5     22.79     48.29
i6     18.19     48.77
i7     22.00     49.65
i8     18.12     51.94
i9     23.13     54.33
i10    19.00     54.87
i11    19.46     56.46
i12    17.83     58.83
i13    32.81     59.13
i14    22.13     60.73
i15    23.46     61.12
i16    16.81     63.10
i17    21.35     65.96
i18    14.87     66.40
i19    33.00     70.42
i20    25.19     70.48
i21    17.77     71.98
i22    22.44     72.00
i23    22.87     72.23
i24    26.52     72.23
i25    21.00     73.44
i26    37.52     74.25
i27    21.69     74.77
i28    27.40     76.33
i29    30.69     81.02
i30    19.56     81.85
i31    30.58     82.56
i32    41.12     83.33
i33    15.38     83.40
i34    17.87     91.81
i35    25.54     91.81
i36    39.00     92.96
i37    20.44     95.17
i38    30.10     101.40
i39    20.90     114.13
i40    48.71     115.46
;
```

²<http://amsterdamoptimization.com/models/statistics/expdata.inc>

1.2. Solving the normal equations. The standard way of formulating the OLS estimators is³

$$(2) \quad \hat{\beta} = (X'X)^{-1}X'y$$

where $\hat{\beta}$ denotes the estimate of β . This implies a simple linear formulation to find $\hat{\beta}$ using the so-called ‘normal equations’:

$$(3) \quad (X'X)\hat{\beta} = X'y$$

This is a system of linear equations. Such a system can be solved with GAMS as an LP using a *dummy objective*. A model that illustrates this is reproduced below. You can verify that it will give the same results.

1.2.1. Model ols2.gms.⁴

```
$ontext
  Ordinary Least Squares (OLS) by solving
  the normal equations.

  Erwin Kalvelagen, october 2000

$offtext

set i /i1*i40/;

$include expdata.inc

set j 'parameters to be estimated' /
      constant 'constant term'
      coeff1 'income coefficient'
      ;
alias (j,jj);

parameters
  X(i,j)   'the X matrix in standard OLS notation (dependent variables)'
  y(i)     'the y vector in standard OLS notation (independent variables)'
  XX(j,jj) "the matrix (X'X)"
  Xy(j)    "the vector (X'y)"
  ;
X(i,'constant') = 1;
X(i,'coeff1')  = data(i,'income');

y(i) = data(i,'expenditure');

XX(j,jj) = sum(i, X(i,j)*X(i,jj));

Xy(j) = sum(i, X(i,j)*y(i));

equations
  dummy_eq  'dummy objective equation'
  normal(j) "normal equations (X'X)b = X'y"
  ;
variables
  b(j)      'parameters to estimate'
  dummy_var 'dummy objective variable'
  ;
dummy_eq.. dummy_var =e= b('constant');
normal(j).. sum(jj, XX(j,jj)*b(jj)) =e= Xy(j);

model ols2 /dummy_eq,normal/;
```

³In this chapter we use x' to denote transposition, i.e. $x' = x^T$.

⁴<http://amsterdamoptimization.com/models/statistics/ols2.gms>

```
| solve ols2 using lp minimizing dummy_var;
| display b.l;
```

Notice that in some of the explanatory text a single quote is used inside the text. This can be done, but the explanatory text need then to be surrounded by double quotes.

It is noted that solving the normal equations is not numerically stable. A better way is to use QR or SVD decomposition. For an example of a QR based least square solver for GAMS see [31].

1.3. OLS Statistics. Many of the other statistics can be found relatively easily. We start with σ^2 or the variance of the estimate. This is calculated as:

$$(4) \quad \sigma^2 = \text{SSE}/(n - k)$$

where $(n - k)$ is the number of degrees of freedom (the number of observations minus the number of estimated parameters) and $\text{SSE} = \sum_i \epsilon_i^2$. The R^2 statistic (coefficient of determination) can easily be calculated using a matrix A ([47]) defined by

$$(5) \quad A = I - \frac{1}{n}(\iota\iota')$$

where ι is a vector of ones. Now:

$$(6) \quad R^2 = 1 - \frac{\text{SSE}}{y'Ay}$$

The adjusted R^2 coefficient (adjusted for the degrees of freedom), denoted by \bar{R}^2 can be written as:

$$(7) \quad \bar{R}^2 = 1 - \frac{n - 1}{n - k}(1 - R^2)$$

The logarithm of likelihood function can be written as:

$$(8) \quad \ln L(\beta, \sigma^2 | y, X) = -\frac{n}{2} \ln 2\pi - \frac{n}{2} \ln \sigma^2 - \frac{(y - X\beta)'(y - X\beta)}{2\sigma^2}$$

which CHAZAM estimates by:

$$(9) \quad \text{llf} = -\frac{n}{2} \ln \left(2\pi \frac{\text{SSE}}{n} \right) - \frac{n}{2}$$

The standard errors of $\hat{\beta}$ can be calculated using

$$(10) \quad \text{Var}(\hat{\beta}) = \sigma^2(X'X)^{-1}$$

which can be solved using a system of linear equations. The standard errors are simply the square roots of the diagonal elements of this variance-covariance matrix. The t -statistic can be calculated simply by dividing β_k by its standard error.

The calculation of the above statistics are implemented in the following model:

1.3.1. Model ols3.gms.⁵

⁵<http://amsterdamoptimization.com/models/statistics/ols3.gms>

```

$ontext
  OLS plus statistics
  Erwin Kalvelagen, november 2000
$offtext

set i /i1*i40/;

$include expdata.inc

set j /'constant','coeff1';
alias (j,jj);
alias (i,ii);

parameters
  X(i,j)   'the X matrix in standard OLS notation (dependent variables)'
  y(i)     'the y vector in standard OLS notation (independent variables)'
  XX(j,jj) "the matrix (X'X)"
  Xy(j)    "the vector (X'y)"
;
X(i,'constant') = 1;
X(i,'coeff1')   = data(i,'income');

y(i) = data(i,'expenditure');

XX(j,jj) = sum(i, X(i,j)*X(i,jj));

Xy(j) = sum(i, X(i,j)*y(i));

equations
  dummy_eq  'dummy objective equation'
  normal(j) "normal equations (X'X)b = X'y"
;
variables
  b(j)      'parameters to estimate'
  dummy_var 'dummy objective variable'
;
dummy_eq.. dummy_var =e= 0;
normal(j).. sum(jj, XX(j,jj)*b(jj)) =e= Xy(j);

model ols2 /dummy_eq,normal/;
solve ols2 using lp minimizing dummy_var;

display "-----estimates-----",
      b.l;

parameters
  residual(i)   'residuals (errors)'
  yhat(i)       'predicted y'
  A(i,i)        "Theil's A matrix: I - (1/n) (iota*iota)"
;
scalars
  sse           'sum of squared errors'
  sst           'total sum of squares'
  ssr           'regression sum of squares'
  n             'number of observations'
  df            'degrees of freedom'
  r2            'r-square'
  r2adj         'r-square adjusted'
  llf           'log of the likelihood function'
  sigma_squared 'variance of the estimate'
  sigma         'standard error of the estimate'
  pi            '3.1415...'
;

```

```

yhat(i) = sum(j, x(i,j)*b.l(j));
residual(i) = y(i) - yhat(i);

sse = sum(i, sqr(residual(i)));

n = card(i);
df = n - 2;
sigma_squared = sse/df;
sigma = sqrt(sigma_squared);
pi = 4*arctan(1);

A(i,ii) = 1$sameas(i,ii) - (1/n);

sst = sum((i,ii), data(i,'expenditure')*A(i,ii)*data(ii,'expenditure'));
ssr = sst - sse;

r2 = 1-sse/sst;

r2adj = 1 - ((n-1)/df)*(1-r2);

llf = -(n/2)*log(2*pi*sse/n)-(n/2);

display "-----statistics-----",
      sse,
      sigma_squared,
      sigma,
      r2,
      r2adj,
      llf;

alias (j,jjj);
variable var(j,jj) 'variance of the estimators';
equation variance(j,jj);
variance(j,jj).. sum(jjj, XX(j,jjj)*var(jjj,jj)) =e= sigma_squared$sameas(j,jj);

model mvar /dummy_eq,variance/;
solve mvar using lp minimizing dummy_var;

parameters
  se(j)      'standard error'
  t(j)       "t ratio's"
  partial(j)
;
se(j) = sqrt(var.l(j,j));
t(j) = b.l(j)/se(j);
partial(j) = t(j)/sqrt(sqr(t(j))+df);

display "-----",
      se,
      t,
      partial
;

```

1.4. Confidence Intervals. The computation of confidence intervals requires the availability of critical values of the Student t distribution. A GAMS include file with a table of those values can be found in [34].

Alternatively for models with a sufficiently large number of degrees of freedom $df = n - k$ we can use a normal approximation:

```

set prob /p1,p2,p3,p4,p5,p6/;
parameter probval(prob) /
  p1 0.10, p2 0.05, p3 0.025, p4 0.01, p5 0.005, p6 0.001

```

```

/;

parameter qnorm(prob) /
  p1 1.281552, p2 1.644854, p3 1.959964, p4 2.326348, p5 2.575829, p6 3.090232
/;
```

Then we can form confidence intervals by:

```

set ival 'confidence interval' /lo,up/;
scalar ndf 'degrees of freedom';
ndf = card(i) - card(k);
scalar alpha 'significance level' /0.025/;
scalar qt 'critical value';

abort$(ndf<30) "Normal approximation not valid";

qt = sum(prob$(probval(prob)=0.025), qnorm(prob));

parameter ols_conf_ival(k,ival);
ols_conf_ival(k,'lo') = beta.l(k) - qt*ols_se(k);
ols_conf_ival(k,'up') = beta.l(k) + qt*ols_se(k);
display ols_conf_ival;
```

where `ols_se(k)` is the standard error of coefficient k .

Note that the solver LS[31] will write the confidence intervals to a GDX file `ls.gdx`. They can be retrieved as follows:

```

*-----*
* read confidence intervals from gdx file
*-----*

sets
  alpha /'90%','95%','97.5%','99%'/;
  names /'c','h','h3'/
  interval /'lo','up'/
;
parameter confint(alpha,names,interval);
execute_load 'ls.gdx',confint;

display confint;
```

This will display:

```

---- 77 PARAMETER confint

          lo           up

90% .c      118.261     174.912
90% .h      -2.637     -1.325
90% .h3 3.758602E-4 4.788719E-4
95% .c      111.959     181.214
95% .h      -2.783     -1.180
95% .h3 3.644011E-4 4.903310E-4
97.5%.c    105.900     187.272
97.5%.h    -2.923     -1.039
97.5%.h3 3.533844E-4 5.013478E-4
99% .c      98.041     195.131
99% .h      -3.105     -0.857
99% .h3 3.390937E-4 5.156384E-4
```

2. LAD: LEAST ABSOLUTE DEVIATION

OLS (Ordinary Least Squares) is based on minimizing a sum of squared errors. A more *robust* estimator can be developed by minimizing the sum of absolute values of errors. This approach gives less weight to outliers. Least squares not only is more sensitive to outliers, but also assumes a Gaussian distribution for the errors. If the

errors have a distribution that is non-Gaussian, e.g. with a fatter tail, estimators based on a LAD norm are sometimes considered more appropriate.

2.1. LAD model formulations. For our linear model $y = X\beta + \epsilon$ the optimization problem becomes

$\text{LAD} \quad \begin{aligned} &\underset{\beta}{\text{minimize}} \quad \sum_i \epsilon_i \\ &\text{subject to} \quad y_i = \sum_j x_{i,j} \beta_j + \epsilon_i \end{aligned}$

The actual implementation is a simple linear programming model using the variable splitting technique: replace ϵ_i by $\epsilon_i^+ - \epsilon_i^-$ and $|\epsilon_i|$ by $\epsilon_i^+ + \epsilon_i^-$ where $\epsilon_i^+, \epsilon_i^- \geq 0$ are non-negative variables. We don't need to add the nonlinear constraint $\epsilon_i^- \epsilon_i^+ = 0$ to enforce that one of the factors $\epsilon_i^+, \epsilon_i^-$ is zero, as the minimization of $\epsilon_i^+ + \epsilon_i^-$ will automatically force this. The resulting LP model is:

$$(11) \quad \begin{aligned} &\min \sum_i \epsilon_i^+ + \epsilon_i^- \\ &y_i = \sum_j x_{i,j} \beta_j + \epsilon_i^+ - \epsilon_i^- \\ &\epsilon_i^+, \epsilon_i^- \geq 0 \end{aligned}$$

As slightly different formulation is:

$$(12) \quad \begin{aligned} &\min \sum_i e_i \\ &y_i = \sum_j x_{i,j} \beta_j + \epsilon_i \\ &e_i \geq \epsilon_i \\ &e_i \geq -\epsilon_i \end{aligned}$$

This implies that $e_i = |\epsilon_i|$.

If the number of constraints is much larger than the number of variables (i.e. the number of observations is much larger than the number of parameters to be estimated), we may think of a few other formulations. First it is noted that we can eliminate the variable ϵ :

$$(13) \quad \begin{aligned} &\min \sum_i e_i \\ &e_i \geq y_i - \sum_j x_{i,j} \beta_j \\ &e_i \geq -y_i + \sum_j x_{i,j} \beta_j \end{aligned}$$




FIGURE 2. OLS and LAD Estimation

Second, we can formulate the dual problem [13]:

$$\begin{aligned}
 & \max \sum_i y_i v_i - \sum_i y_i w_i \\
 & v_i + w_i = 1 \\
 (14) \quad & \sum_i x_{i,j} v_i - \sum_i x_{i,j} w_i = 0 \\
 & v_i, w_i \geq 0
 \end{aligned}$$

We can now substitute out $w_i = 1 - v_i$, resulting in:

$$\begin{aligned}
 & \max - \sum_i y_i + 2 \sum_i y_i v_i \\
 (15) \quad & - \sum_i x_{i,j} + 2 \sum_i x_{i,j} v_i = 0 \\
 & 0 \leq v_i \leq 1
 \end{aligned}$$

which can be simplified to:

$$\begin{aligned}
 & \max \sum_i y_i v_i \\
 (16) \quad & \sum_i x_{i,j} v_i = \frac{1}{2} \sum_i x_{i,j} \\
 & 0 \leq v_i \leq 1
 \end{aligned}$$

LAD regression is also known as MAD (Minimum Absolute Deviations) regression, LAV (Least Absolute Value) regression and ℓ_1 norm estimation. Interestingly, LAD has older traces back into history than least squares fitting. [9, 8] mention that the earliest references to a curve fitting criterion based on least absolute deviations are by Boscovitch⁶, formulated somewhere around 1757, while the famous Legendre did publish his “Principle of Least Squares” as recent as 1805.

For some the of statistical properties of LAD estimators see the review [15].

2.1.1. Model lad.gms.⁷

This model finds LAD estimates directly by minimizing $\sum_i |\epsilon_i|$.

```
$ontext
  Least Absolute Deviation (LAD).
  Erwin Kalvelagen, october 2000

$offtext

set i /i1*i40/;

$include expdata.inc

variables
  constant      'estimate constant term coefficient'
  income        'estimate income coefficient'
  sad           'sum of absolute deviations'
;

positive variables
  res_plus(i)   'error term (plus term)'
  res_min(i)    'error term (minus term)'
;

equations
  fit(i)        'the linear model'
  obj          'objective'
;

obj..    sad == sum(i, res_plus(i) + res_min(i));
fit(i).. data(i,'expenditure') == constant + income*data(i,'income')
         + res_plus(i) - res_min(i);

model ols1 /obj,fit/;

solve ols1 minimizing sad using lp;
display constant.l, income.l, sad.l;
```

2.1.2. Model lad2.gms.⁸

This is an alternative formulation without variable splitting.

```
$ontext
  Least Absolute Deviation (LAD), alternative formulation.
```

⁶Roger Joseph Boscovitch, also spelled as Rudjer J. Bōsković (1711–1787), a Jesuit priest and prominent scientist, who spent most of his life in Rome, was born in Ragusa (now called Dubrovnik). For a fascinating account on his work on estimating the length of a meridian arc near Rome, see [46].

⁷<http://amsterdamoptimization.com/models/statistics/lad.gms>

⁸<http://amsterdamoptimization.com/models/statistics/lad2.gms>

```

Erwin Kalvelagen, october 2000

$offtext

set i /i1*i40/;

$include expdata.inc

variables
  constant    'estimate constant term coefficient'
  income      'estimate income coefficient'
  sad         'sum of absolute deviations'
  e(i)        'error term'
  abse(i)     'absolute error term'
;

positive variables
  res_plus(i)  'error term (plus term)'
  res_min(i)   'error term (minus term)'
;

equations
  fit(i)      'the linear model'
  plus(i)     'plus inequalities'
  min(i)      'min inequalities'
  obj        'objective'
;

obj..      sad =e= sum(i, abse(i));
fit(i)..  data(i,'expenditure') =e= constant + income*data(i,'income') + e(i);
plus(i)..  abse(i) =g= e(i);
min(i)..  abse(i) =g= -e(i);

model lad2 /obj,fit,plus,min/;

solve lad2 minimizing sad using lp;
display constant.l, income.l, sad.l;

```

2.2. Best subset LAD regression. An interesting extension of the LAD regression model is to find the best the best subset of variables to include in the regression [1]. Suppose we want k out of a possible m independent variables in the regression equation. The reasons for restricting the number of variables can include making the equation easier to understand, or reducing (future) cost in data collection, analysis and interpretation.

Of course we can run all possible combinations, but there are $\frac{m!}{k!(m-k)!}$ possible ways of choosing k out of m . A first model for this problem could read as:

$$\begin{aligned}
 & \min \sum_i |\epsilon_i| \\
 & y_i = \sum_j x_{i,j} \beta_j \delta_j + \epsilon_i \\
 & \sum_j \delta_j = k \\
 & \delta_j \in \{0, 1\}
 \end{aligned}
 \tag{17}$$

where δ_j are binary variables. This non-linear formulation can be transformed into a linear one:

$$\begin{aligned}
 & \min \sum_i |\epsilon_i| \\
 & y_i = \sum_j x_{i,j} \beta_j + \epsilon_i \\
 (18) \quad & \sum_j \delta_j = k \\
 & \beta_j \leq M \delta_j \\
 & \beta_j \geq -M \delta_j \\
 & \delta_j \in \{0, 1\}
 \end{aligned}$$

where M is a constant that need to be chosen with some care. It should be large enough so that $\beta_j \leq M \delta_j$ and $\beta_j \geq -M \delta_j$ are non-binding for $\delta_j = 1$. I.e. M is a bound on β_j . On the other hand too large values lead to (numerical) problems in the MIP solver.

2.2.1. Model subsetlad.gms.⁹

```

$ontext
    Best subset LAD regression
    Erwin Kalvelagen, august 2001
$offtext

sets
    i /i1*i250/
    j /j1*j10/
;

parameter x(i,j) 'data independent variables';
parameter y(i)   'data dependent variable';

x(i,j) = normal(0,10);
y(i) = sum(j, x(i,j)*ord(j)) + normal(0,1);

variable b(j)      'parameters to estimate';
binary variable delta(j) 'subset selection';
variable z         'sum of absolute deviations';
positive variable ep(i)  'positive deviations';
positive variable em(i)  'negative deviations';

equations
    obj      'objective'
    fit(i)   'equation to be fitted'
    subset   'select k'
    bigm1(j) 'big M formulation'
    bigm2(j) 'big M formulation'
;

scalar M 'big-M' /100/;

scalar k 'number of variables to select' /3/;

obj..      z =e= sum(i, ep(i)+em(i));
fit(i)..   y(i) =e= sum(j, x(i,j)*b(j)) + ep(i) - em(i);

```

⁹<http://amsterdamoptimization.com/models/statistics/subsetlad.gms>

```

bigm1(j).. b(j) =l= M*delta(j);
bigm2(j).. b(j) =g= -M*delta(j);
subset.. sum(j, delta(j)) =e= k;

option optcr=0;
option iterlim=1000000;
model ksubset /all/;
solve ksubset using mip minimizing z;

```

Not all solver have an easy time on this model. BDMLP for instance needs 189 nodes to solve this model to optimality, while a complete enumeration would take 120 nodes. The reason is that there are 10 binary variables, which leads to a theoretical worst case of $2^{10} = 1024$ nodes.

In [1] a special purpose branch-and-bound code is developed for this problem.

2.3. Trend breaks in LAD regression. The estimation of structural breaks has become a popular technique in econometrics [5, 6]. Estimating a trend break in a LAD regression problem can be done by a grid search as suggested in [4]. However, we can also do this directly using a MIP model.




FIGURE 3. LAD Estimation of a trend break

The model can be developed as follows. First we have two regression equations:

$$(19) \quad \begin{aligned} y_i &= a^{(1)} + b^{(1)}x_i + e_i^{(1)} \\ y_i &= a^{(2)} + b^{(2)}x_i + e_i^{(2)} \end{aligned}$$

We introduce binary variables to determine in which segment each x_i is:

$$(20) \quad \begin{aligned} x_i &\leq B + \delta_i M_1 \\ x_i &\geq B - (1 - \delta_i) M_1 \end{aligned}$$

where B is the variable indicating the location of the break. The constant M_1 needs to be chosen as small as possible. The binary variables δ_i are also used to determine which error term contributes to the objective:

$$(21) \quad \eta_i = \begin{cases} e_i^{(1)} & \text{if } \delta_i = 0 \\ e_i^{(2)} & \text{if } \delta_i = 1 \end{cases}$$

After applying variable splitting:

$$(22) \quad \begin{aligned} \eta_i &= \eta_i^+ - \eta_i^- \\ \eta_i^+, \eta_i^- &\geq 0 \end{aligned}$$

we can form the objective:

$$(23) \quad \min \sum_i \eta_i^+ + \eta_i^-$$

Equation (21), combined with (22) can be written as a set of inequalities:

$$(24) \quad \begin{aligned} e_i^{(1)} - \delta_i M_2 &\leq \eta_i^+ - \eta_i^- \leq e_i^{(1)} + \delta_i M_2 \\ e_i^{(2)} - (1 - \delta_i) M_2 &\leq \eta_i^+ - \eta_i^- \leq e_i^{(2)} + (1 - \delta_i) M_2 \end{aligned}$$

Again M_2 should be chosen as small as possible. The actual choice of M_2 is not straightforward. It depends on the angle in the kink.

The following GAMS model demonstrates the procedure.

2.3.1. Model *l1break.gms*. ¹⁰

```
$ontext
  L1 regression with a trend break
  Erwin Kalvelagen, oct 2003

$offtext

set i 'number of observations' /i1*i50/;

parameter x(i);
parameter y(i);

*
* generate test data
* x must be ordered: x(i) >= x(i-1).
*
set i1(i) /i1*i30/;
set i2(i) /i31*i50/;

x(i) = 0;
loop(i,
  x(i) = x(i-1) + uniform(0.1,10);
);
display x;

y(i1) = 200 - 2*x(i1) + normal(0,10);
y(i2) = -220 + x(i2) + normal(0,10);
```

¹⁰<http://amsterdamoptimization.com/models/statistics/l1break.gms>

```

display y;

*
* coefficients to estimate
*
variables a1,b1,a2,b2,break;

*
* error terms
*
variables e1(i),e2(i);

*
* equations to fit
*
equations fit1(i),fit2(i);
fit1(i).. y(i) =e= a1 + b1*x(i) + e1(i);
fit2(i).. y(i) =e= a2 + b2*x(i) + e2(i);

*
* we use binary variables b(i) to indicate
* if x(i) is in segment 1 or segment 2.
*
binary variable b(i);

equations seg1(i),seg2(i);
scalar m1; m1 = smax(i,x(i))-smin(i,x(i));

seg1(i).. x(i) =l= break + b(i)*M1;
seg2(i).. x(i) =g= break - (1-b(i))*M1;

variable u(i) 'either e1(i) or e2(i)';
equation udef1(i),udef2(i),udef3(i),udef4(i);

*
* 10*range of y is generous for m2
*
scalar m2;
m2 = 10*[smax(i,y(i))-smin(i,y(i))];

udef1(i).. u(i) =g= e1(i) - b(i)*m2;
udef2(i).. u(i) =l= e1(i) + b(i)*m2;
udef3(i).. u(i) =g= e2(i) - (1-b(i))*m2;
udef4(i).. u(i) =l= e2(i) + (1-b(i))*m2;

*
* variable splitting
*
positive variable v(i),w(i);
equation split(i);
split(i).. v(i) - w(i) =e= u(i);

*
* objective
*
variable l1;
equation obj;

obj.. l1 =e= sum(i, v(i) + w(i));

model m/all/;
option optcr=0;
option mip=cplex;
solve m minimizing l1 using mip;

parameter res(i,*);
res(i,'x') = x(i);
res(i,'b') = b.l(i);
res(i,'u') = u.l(i);

```

```

| res(i,'e1') = e1.l(i);
| res(i,'e2') = e2.l(i);

display break.l,res;

*
* produce some graphs
*
file dat0 /10.dat/;
file dat1 /11.dat/;
file dat2 /12.dat/;
loop(i,
  put dat0,x(i):12:7, y(i):12:7/
  if (b.l(i)<0.5,
    put dat1,x(i):12:7, (y(i)-e1.l(i)):12:7/
  else
    put dat2,x(i):12:7, (y(i)-e2.l(i)):12:7/
  );
);
putclose dat0;
putclose dat1;
putclose dat2;

file plt /11.plt/;
putclose plt,
'plot "10.dat" title "data",
' "11.dat" title "segment 1" with lines,
' "12.dat" title "segment 2" with lines'
' pause -1'
execute 'gnuplot 11.plt';

```

3. NONLINEAR LEAST SQUARES

3.1. Nonlinear regression using GAMS. Minimizing the sum of squared errors for nonlinear relationships is a task that an NLP solver is well equipped to do. The basic regression model is:

$$(25) \quad y = f_{\theta}(X) + \epsilon$$

where y and X form the observations for the dependent and independent variables, and θ is the vector of parameters to be estimated. In the optimization model this becomes:

$$(26) \quad \min_{\theta} \|y - f_{\theta}(X)\|_2^2$$

with θ becoming a decision variable and y, X are data.

3.2. Fitting a CES production function. As an example consider a CES (Constant Elasticity of Substitution) production function, an important equation used in many economic models. A search in the GAMS model library shows a handful of models that have CES functions. A production function $Q = f(K, L)$ measures output given inputs consisting of the ‘factors of production’ (in our case we have 2 factors: labor L and capital K). A simple production function often used in the economic literature is the Cobb-Douglas production function [50]. It looks like:

$$(27) \quad Q = \lambda K^{\alpha} L^{\beta}$$

A more complicated function that is very well known is the CES production function. CES functions were introduced by [3], and are also called ACMS functions, after the authors. For more information on CES functions and their limitations in




FIGURE 4. Scatter plots of the CES data set

production theory see [30, 17]. The functional form of a CES production function is:

$$(28) \quad Q = \gamma [\delta L^{-\rho} + (1 - \delta)K^{-\rho}]^{-\frac{\eta}{\rho}}$$

where L is labor, K is capital, Q is output. γ is called the ‘efficiency parameter’ ($\gamma > 0$), δ is the ‘distribution parameter’ ($0 < \delta < 1$), and ρ is the ‘substitution parameter’ ($-1 \leq \rho \leq \infty$). η denotes the degree of homogeneity of the function.

Takings logs, and renaming some parameters, we can write this as:

$$(29) \quad \ln Q = \gamma - \frac{\eta}{\rho} \ln (\delta L^{-\rho} + (1 - \delta)K^{-\rho})$$

As an aside it is noted that CES functions also have an application in Linear Programming theory. Interior point methods are often based on variants of a logarithmic barrier function. However, one can also devise an interior point algorithm for linear programming based on a CES function [36].

A data set from [25] is used as an example for our nonlinear regression problem. It is reproduced in table 2

The optimization model to be solved can be simply stated as:

NLREG minimize $\sum_i r_i^2$ subject to $\ln Q_i = \gamma - \frac{\eta}{\rho} \ln (\delta L_i^{-\rho} + (1 - \delta)K_i^{-\rho}) + r_i$
--




FIGURE 6. Surface of CES production function

VARIABLE eta.L	=	1.013 homogeneity parameter
VARIABLE sse.L	=	1.761 sum of squared errors

These figures are confirmed by the results of the CHAZAM [51] run reproduced in the next section.

3.2.1. Model nls.gms. ¹¹

```
$ontext
Nonlinear least squares.

Example: Estimation of a CES production function

Data set: Table 22.4, page 724 of Griffiths, Hill and Judge,
LEARNING AND PRACTICING ECONOMETRICS, Wiley, 1993.

Erwin Kalvelagen, 2000

$offtext

set i 'observations' /i1*i30/;
set j 'parameters' /L,K,Q/;

table data(i,j)
      L          K          Q
i1    0.228    0.802    0.256918
i2    0.258    0.249    0.183599
i3    0.821    0.771    1.212883
i4    0.767    0.511    0.522568
i5    0.495    0.758    0.847894
i6    0.487    0.425    0.763379
```

¹¹<http://amsterdamoptimization.com/models/statistics/nls.gms>

```

i7    0.678    0.452    0.623130
i8    0.748    0.817    1.031485
i9    0.727    0.845    0.569498
i10   0.695    0.958    0.882497
i11   0.458    0.084    0.108827
i12   0.981    0.021    0.026437
i13   0.002    0.295    0.003750
i14   0.429    0.277    0.461626
i15   0.231    0.546    0.268474
i16   0.664    0.129    0.186747
i17   0.631    0.017    0.020671
i18   0.059    0.906    0.100159
i19   0.811    0.223    0.252334
i20   0.758    0.145    0.103312
i21   0.050    0.161    0.078945
i22   0.823    0.006    0.005799
i23   0.483    0.836    0.723250
i24   0.682    0.521    0.776468
i25   0.116    0.930    0.216536
i26   0.440    0.495    0.541182
i27   0.456    0.185    0.316320
i28   0.342    0.092    0.123811
i29   0.358    0.485    0.386354
i30   0.162    0.934    0.279431
;

parameters
  L(i)      'labor'
  K(i)      'capital'
  Q(i)      'output'
;
L(i) = data(i,'L');
K(i) = data(i,'K');
Q(i) = data(i,'Q');

variables
  gamma      'log of efficiency parameter'
  delta       'distribution parameter'
  rho        'substitution parameter'
  eta         'homogeneity parameter'
  residual(i) 'error term'
  sse        'sum of squared errors'
;
equations
  fit(i)     'the nonlinear model'
  obj        'objective'
;
obj..    sse =e= sum(i, sqr(residual(i)));
fit(i).. log(Q(i)) =e=
           gamma - (eta/rho)*log[delta*L(i)**(-rho) + (1-delta)*K(i)**(-rho)]
           + residual(i);

* initial values
rho.l=1;
delta.l=0.5;
gamma.l=1;
eta.l=1;

model nls /obj,fit/;

solve nls minimizing sse using nlp;
display gamma.l, delta.l, rho.l, eta.l, sse.l;

```

3.2.2. *Output of CHAZAM.* This run is used to verify the solution.

```
*****
Hello/Bonjour/Aloha/Howdy/G Day/Kia Ora/Konnichiwa/Buenos Dias/Nee Hau/Ciao
Welcome to SHAZAM - Version 9.0 - OCT 2000 SYSTEM=LINUX PAR= 781
|_* NONLINEAR LEAST SQUARES AND TESTING FOR AUTOCORRELATED ERRORS
|_*
|_* Example: Estimation of a CES production function
|_*
|_* Data set: Table 22.4, page 724 of Griffiths, Hill and Judge,
|_* LEARNING AND PRACTICING ECONOMETRICS, Wiley, 1993.
|_*
|_* SAMPLE 1 30
|_* READ L K Q
 3 VARIABLES AND      30 OBSERVATIONS STARTING AT OBS      1

|_* GENR LOGQ=LOG(Q)
|_* Estimate the CES production function

|_* NL 1 / NCOEF=4 PCOV ZMATRIX=Z COEF=BETA PREDICT=YHAT
...NOTE..SAMPLE RANGE SET TO:    1,    30
|_* EQ LOGQ=GAMMA-(ETA/RHO)*LOG(DELTA*L**(-RHO)+(1-DELTA)*K**(-RHO))
|_* COEF RHO 1 DELTA .5 GAMMA 1 ETA 1
 3 VARIABLES IN 1 EQUATIONS WITH 4 COEFFICIENTS
 30 OBSERVATIONS

REQUIRED MEMORY IS PAR=      22 CURRENT PAR=      781

COEFFICIENT STARTING VALUES
GAMMA      1.0000      ETA      1.0000      RHO      1.0000
DELTA      0.50000
100 MAXIMUM ITERATIONS, CONVERGENCE =  0.100000E-04

INITIAL STATISTICS :

TIME =      0.000 SEC.  ITER. NO.      0  FUNCT. EVALUATIONS      1
LOG-LIKELIHOOD FUNCTION= -45.75315
COEFFICIENTS
 1.0000000      1.0000000      1.0000000      0.5000000
GRADIENT
 -25.82924      42.55772      5.501555      -6.222794

INTERMEDIATE STATISTICS :

TIME =      0.040 SEC.  ITER. NO.     15  FUNCT. EVALUATIONS     25
LOG-LIKELIHOOD FUNCTION= -0.6460435E-01
COEFFICIENTS
 0.1246125      1.018006      2.750354      0.3581035
GRADIENT
 -0.5488362      0.2883618      0.2721726E-01      -2.287144

FINAL STATISTICS :

TIME =      0.060 SEC.  ITER. NO.     25  FUNCT. EVALUATIONS     35
LOG-LIKELIHOOD FUNCTION= -0.3907423E-01
COEFFICIENTS
 0.1244913      1.012594      3.010941      0.3366735
GRADIENT
 -0.4790552E-03 -0.7407311E-04      0.1256401E-05      -0.1297012E-03
ASYMPTOTIC COVARIANCE MATRIX
GAMMA      0.47917E-02
ETA       0.61486E-03      0.21055E-02
RHO       0.50037E-01      -0.67552E-01      4.0663
DELTA     -0.15592E-02      0.20209E-02      -0.13216      0.97548E-02
          GAMMA      ETA      RHO      DELTA

MAXIMUM LIKELIHOOD ESTIMATE OF SIGMA-SQUARED =  0.58703E-01
SUM OF SQUARED ERRORS =  1.7611
GTRANSPOSE*INVERSE(H)*G  STATISTIC - =  0.11658E-08

COEFFICIENT      ST. ERROR      T-RATIO
GAMMA      0.12449      0.69222E-01      1.7984
```

ETA	1.0126	0.45886E-01	22.068
RHO	3.0109	2.0165	1.4931
DELTA	0.33667	0.98766E-01	3.4088
_ END			
_ STOP			

An alternative formulation for the nonlinear least squares problem is:

$$(30) \quad \begin{aligned} & \min \sum_i \epsilon_i^2 \\ & -\epsilon_i \leq f_\theta(X_i) - y_i \leq \epsilon_i \end{aligned}$$

It is noted that we can easily add inequalities and bounds that are known from theory, and bounds that keep values in a reasonable range:

```
delta.lo = 0.0001;
delta.up = 0.9999;
rho.lo = -0.9999;
* rho.lo = 0.0001; would be even better
rho.up = 100;
```

This is likely more powerful than found in the non-linear regression found in many statistical packages.

3.3. Nonlinear regression using NLS. A specialized Non-linear Regression solver called NLS[32] is available. It has a built-in nonlinear least squares solver NL2SOL[29]. It provides numerous statistical output including the standard errors and the variance-covariance matrix. In addition it can take a starting point as calculated by a GAMS NLP solver such as MINOS or CONOPT.

The above model can be specified as:

```
variables
  gamma      'log of efficiency parameter'
  delta       'distribution parameter'
  rho         'substitution parameter'
  eta         'homogeneity parameter'
  residual(i) 'error term'
  sse         'sum of squared errors'
;

equations
  fit(i)     'the nonlinear model'
  obj        'objective'
;

obj..    sse =e= sum(i, sqr(residual(i)));
fit(i).. log(Q(i)) =e=
          gamma - (eta/rho)*log[delta*L(i)**(-rho) + (1-delta)*K(i)**(-rho)]
          + residual(i);

* initial values
rho.l=1;
delta.l=0.5;
gamma.l=1;
eta.l=1;

model nls /obj,fit/;
solve nls minimizing sse using nlp;
display gamma.l, delta.l, rho.l, eta.l, sse.l;
```

The output looks like:

```
=====
Nonlinear Least Square Solver V1
```

```

Erwin Kalvelagen, Amsterdam Optimization Modeling Group
www.amsterdamoptimization.com
=====
Nonlinear Least Square Solver: NL2SOL

Parameter      Estimate    Std. Error      t value   Pr(>|t|)
gamma     -1.24491E-01  7.83443E-02  -1.58902E+00  1.24143E-01
delta      3.36673E-01  1.36112E-01   2.47350E+00  2.02326E-02 *
rho       3.01094E+00  2.32337E+00   1.29593E+00  2.06385E-01
eta       1.01259E+00  5.06832E-02   1.99789E+01  2.66997E-17 ***
---
Signif. codes:  0 ‘***’ 0.001 ‘**’ 0.01 ‘*’ 0.05 ‘.’ 0.1 ‘ ’ 1

Residual standard error:  2.60257E-01   on 26 degrees of freedom

DLL version: _GAMS_GDX_237_2007-01-09
GDX file: nls.gdx

```

4. MAXIMUM LIKELIHOOD ESTIMATION

4.1. Maximum likelihood estimation of the Gamma distribution. Consider data collected on times between failures of air conditioning units in different aircraft[14]. We assume the times between failures are independent random variables with a Gamma distribution. Given a mean time between failures μ and a shape parameter β , the density function of the gamma distribution is[48]:

$$(31) \quad f(x) = \frac{(\beta/\mu)(\beta x/\mu)^{\beta-1} e^{-\beta x/\mu}}{\Gamma(\beta)}$$

The log likelihood function can now be written as:

$$(32) \quad L(\mu, \beta) = n [\ln \beta - \ln \mu - \ln \Gamma(\beta)] + \sum_{i=1}^n (\beta - 1) \ln \left(\frac{\beta x_i}{\mu} \right) - \sum_{i=1}^n \frac{\beta x_i}{\mu}$$

We can maximize this function using the `loggamma(.)` function.

The method of moments estimator of β is

$$(33) \quad \hat{\beta} = \left(\frac{\hat{\mu}}{\hat{\sigma}} \right)^2$$

which can be used as an (excellent) initial point for the optimization problem.

4.1.1. Model *mlgamma.gms*.¹²

```

$ontext
Maximum Likelihood estimation of parameters of the gamma distribution

Erwin Kalvelagen, april 2004.

Data from:
COX, D. R. AND SNELL, E. J., (1981)
Applied Statistics: Principles and Examples,
London: Chapman and Hall.

Example from:
Luke Tierney, July 1989
XLISP-STAT, A Statistical Environment Based on the XLISP Language (Version 2.0)
Technical Report Number 528, University of Minnesota, School of Statistics

```

¹²<http://amsterdamoptimization.com/models/statistics/mlgamma.gms>

```
$offtext

set i 'observations' /i1*i29/

parameter x(i) 'times (in operating hours) between failures of airco units on several aircraft'
/
  i1  90,  i2  10,  i3  60,  i4  186,  i5  61
  i6  49,  i7  14,  i8  24,  i9  56,  i10 20
  i11 79,  i12 84,  i13 44,  i14 59,  i15 29
  i16 118,  i17 25,  i18 156,  i19 310,  i20 76
  i21 26,  i22 44,  i23 23,  i24 62,  i25 130
  i26 208,  i27 70,  i28 101,  i29 208
/;

scalar n;
n = card(i);

scalar average;
average = sum(i, x(i))/n;

scalar stdev 'standard deviation';
stdev = sqrt(sum(i, sqr(x(i)-average))/(n-1));

display average,stdev;

variables beta,mu,like;
equations loglike;

loglike.. like =e= n*[log(beta)-log(mu)-loggamma(beta)] +
  sum(i, (beta-1)*log(beta*x(i)/mu)) -
  sum(i, beta*x(i)/mu);

*
* lowerbounds so log() and loggamma() are safe
*
beta.lo = 0.0001;
mu.lo = 0.0001;

*
* initial values using moments estimates
*
mu.l = average;
beta.l = sqr(average/stdev);

model m /loglike/;
solve m using nlp maximimizing like;
```

The resulting estimates for the parameters μ and β are:

	LOWER	LEVEL	UPPER	MARGINAL
---- VAR beta	0.0001	1.6710	+INF	1.004352E-11
---- VAR mu	0.0001	83.5172	+INF	1.944001E-13
---- VAR like	-INF	-155.3468	+INF	.

4.2. Maximum likelihood estimation of the Beta distribution. The log likelihood function of the beta distribution with parameters α and β is:

(34)

$$\ln L = n [\ln \Gamma(\alpha + \beta) - \ln \Gamma(\alpha) - \ln \Gamma(\beta)] + \sum_{i=1}^n (\alpha - 1) \ln(x_i) + \sum_{i=1}^n (\beta - 1) \ln(1 - x_i)$$

This function can be implemented straightforwardly using the `loggamma` function.

The first two moments of the beta distribution, lead to two equations in two variables defining the method of moments estimator of α and β :

$$(35) \quad E(X) = \frac{\alpha}{\alpha + \beta}$$

$$Var(X) = \frac{\alpha\beta}{(\alpha + \beta)^2(\alpha + \beta + 1)}$$

or

$$(36) \quad \hat{\alpha} = \left[\frac{\hat{\mu}(1 - \hat{\mu})}{\hat{\sigma}^2} - 1 \right] \hat{\mu}$$

$$\hat{\beta} = \left[\frac{\hat{\mu}(1 - \hat{\mu})}{\hat{\sigma}^2} - 1 \right] (1 - \hat{\mu})$$

4.2.1. Model *mlbeta.gms*. ¹³

```
$ontext
Fitting of beta distribution through maximum likelihood
Erwin Kalvelagen, april 2004

Reference:
Johnson, Kotz, and Balakrishnan, (1994),
Continuous Univariate Distributions, Volumes I and II,
2nd. Ed., John Wiley and Sons.

$offtext

set i 'cases' /i1*i75/;
parameter x(i) /
i1 4.973016e-01, i2 3.558841e-01, i3 2.419578e-02, i4 1.913753e-01, i5 4.919495e-01
i6 9.790016e-01, i7 3.856570e-01, i8 1.568263e-01, i9 8.040481e-01, i10 8.108720e-01
i11 6.016693e-01, i12 3.691279e-02, i13 9.454942e-01, i14 1.853702e-01, i15 3.496894e-01
i16 4.249933e-01, i17 9.900851e-01, i18 6.308701e-01, i19 4.474022e-02, i20 4.408432e-03
i21 3.718974e-03, i22 1.066217e-01, i23 5.304127e-01, i24 6.781648e-01, i25 6.206926e-02
i26 4.048511e-01, i27 4.941163e-01, i28 1.644695e-01, i29 2.285463e-02, i30 5.654344e-05
i31 2.657641e-01, i32 7.316988e-01, i33 6.789551e-01, i34 3.624824e-01, i35 7.429815e-03
i36 1.503384e-01, i37 7.314336e-01, i38 4.586442e-02, i39 4.060616e-02, i40 3.395101e-01
i41 9.269645e-01, i42 2.192909e-03, i43 2.511850e-02, i44 4.152490e-01, i45 1.612197e-01
i46 1.512879e-02, i47 1.381864e-01, i48 5.730967e-03, i49 1.185086e-01, i50 7.411310e-01
i51 1.564168e-02, i52 2.206906e-01, i53 9.836009e-01, i54 4.632388e-01, i55 9.968135e-01
i56 8.792355e-04, i57 9.692757e-01, i58 9.823214e-01, i59 1.248862e-01, i60 1.598848e-01
i61 9.561613e-02, i62 2.513807e-01, i63 4.435097e-01, i64 8.852468e-01, i65 1.149253e-02
i66 6.575999e-01, i67 8.236305e-01, i68 7.388426e-01, i69 6.382491e-01, i70 3.426699e-01
i71 1.244351e-01, i72 2.753017e-05, i73 1.625740e-01, i74 2.953334e-02, i75 8.739085e-02
;

scalar n;
n = card(i);

scalar average;
average = sum(i, x(i))/n;

scalar stdev 'standard deviation';
stdev = sqrt(sum(i, sqr(x(i)-average))/(n-1));

variables alpha,beta,like;
equations loglike;

loglike.. like =e= n*[loggamma(alpha+beta)-loggamma(alpha)-loggamma(beta)] +
sum(i, (alpha-1)*log(x(i))) +
sum(i, (beta-1)*log(1-x(i)));



```

¹³<http://amsterdamoptimization.com/models/statistics/mlbeta.gms>

```

*
* lowerbounds so log() is safe
*
alpha.lo = 0.0001;
beta.lo = 0.0001;

*
* initial values using matching moments estimates
*
scalar tmp;
tmp = average*(1-average)/sqr(stdev) - 1;
alpha.l = tmp*average;
beta.l = tmp*(1-average);

display alpha.l,beta.l;

model m /loglike/;
solve m using nlp maximizing like;

display alpha.l,beta.l;

```

4.3. The Tobit model. The Tobit model[49] deals with a dependent variable y which is only observed if $y_i > 0$. To be more precise, in the regression equation

$$(37) \quad y' = X\beta + \varepsilon$$

we only observe

$$(38) \quad y = \max\{y', 0\}$$

A well-known method to estimate β in this case is to maximize the log-likelihood function[24, 7]:

$$(39) \quad \begin{aligned} \ln L &= \sum_{y_i > 0} -\frac{1}{2} \left[\ln(2\pi) + \ln \sigma^2 + \frac{(y_i - x'_i \beta)^2}{\sigma^2} \right] + \sum_{y_i=0} \ln \left[1 - \Phi \left(\frac{x'_i \beta}{\sigma} \right) \right] \\ &= \sum_{y_i > 0} -\frac{1}{2} \left[\ln \sigma^2 + \frac{(y_i - x'_i \beta)^2}{\sigma^2} \right] + \sum_{y_i=0} \ln \left[1 - \Phi \left(\frac{x'_i \beta}{\sigma} \right) \right] - \sum_{y_i > 0} \ln \sqrt{2\pi} \end{aligned}$$

where $\Phi(\cdot)$ is the distribution function of the standard normal distribution $N(0, 1)$.

We can use OLS estimates as initial values for the nonlinear optimization problem. These estimates have been calculated using the LS solver from [31].

4.3.1. Model tobit.gms.¹⁴

```

$ontext

Tobit analysis.

Use least squares solution as starting point for max likelihood optimization.

Erwin Kalvelagen, dec 2004

References:
    William H. Greene, "Econometric Analysis", 5th ed.

Data from Fair (1977).

$offtext

```

¹⁴<http://amsterdamoptimization.com/models/statistics/tobit.gms>


```

1669 2 1. 0 27.0 4.000 0 1 17. 4.0 3 1 2. 0. 1.
1674 7 1. 0 17.5 0.750 1 2 12. 7.5 3 5 12. 0. 1.
1682 4 1. 0 32.0 15.000 1 5 18. 12.5 5 4 7. 0. 1.
1685 4 1. 0 22.0 4.000 0 1 16. 7.5 3 5 7. 0. 1.
1697 2 1. 1 32.0 4.000 1 4 18. 20.0 6 4 2. 0. 1.
1716 1 1. 0 22.0 1.500 1 3 18. 20.0 5 2 1. 0. 1.
1730 3 1. 0 42.0 15.000 1 2 17. 40.0 5 4 3. 0. 1.
1731 1 1. 1 32.0 7.000 1 4 16. 12.5 4 4 1. 0. 1.
1732 5 1. 1 37.0 15.000 0 3 14. 20.0 6 2 12. 0. 1.
1743 1 1. 1 42.0 15.000 1 3 16. 40.0 6 3 1. 0. 1.
1751 1 1. 1 27.0 4.000 1 1 18. 7.5 5 4 1. 0. 1.
1757 2 1. 1 37.0 15.000 1 4 20. 40.0 7 3 2. 0. 1.
1763 4 1. 1 37.0 15.000 1 3 20. 40.0 6 4 7. 0. 1.
1766 3 1. 1 22.0 1.500 0 2 12. 12.5 3 3 3. 0. 1.
1772 3 1. 1 32.0 4.000 1 3 20. 20.0 6 2 3. 0. 1.
1776 2 1. 1 32.0 15.000 1 5 20. 20.0 6 5 2. 0. 1.
1782 5 1. 0 52.0 15.000 1 1 18. 40.0 5 5 12. 0. 1.
1784 5 1. 1 47.0 15.000 0 1 18. 40.0 6 5 12. 0. 1.
1791 3 1. 0 32.0 15.000 1 4 16. 12.5 4 4 3. 0. 1.
1831 4 1. 0 32.0 15.000 1 3 14. 12.5 3 2 7. 0. 1.
1840 4 1. 0 27.0 7.000 1 4 16. 20.0 1 2 7. 0. 1.
1844 5 1. 1 42.0 15.000 1 3 18. 12.5 6 2 12. 0. 1.
1856 4 1. 0 42.0 15.000 1 2 14. 12.5 3 2 7. 0. 1.
1876 5 1. 1 27.0 7.000 1 2 17. 12.5 5 4 12. 0. 1.
1929 3 1. 1 32.0 10.000 1 4 14. 7.5 4 3 3. 0. 1.
1935 4 1. 1 47.0 15.000 1 3 16. 20.0 4 2 7. 0. 1.
1938 1 1. 1 22.0 1.500 1 1 12. 7.5 2 5 1. 0. 1.
1941 4 1. 0 32.0 10.000 1 2 18. 7.5 5 4 7. 0. 1.
1954 2 1. 1 32.0 10.000 1 2 17. 20.0 6 5 2. 0. 1.
1959 2 1. 1 22.0 7.000 1 3 18. 20.0 6 2 2. 0. 1.
9010 1 1. 0 32.0 15.000 1 3 14. 40.0 1 5 1. 0. 1.

;

*
* create a subset of id's: only of those that are actually used
* in the data set
*

set i(id) 'used records';
i(id)$sum(v$data(id,v),1) = yes;
*display i;

*
* add constant term
*
data(i,'const') = 1;

*
* sanity check. this should be 601
*
scalar n;
n=card(i);
display n;

-----
* OLS regression
-----

set j(v) 'independent variables' /const,z2,z3,z5,z7,z8/;

*
* set up parameters X, y for easier manipulations further on
*
parameter y(id),x(id,j);
y(i) = data(i,'y');
x(i,j) = data(i,j);

variables coeff(j);
variable sse 'sum of squared errors';

```

```

equations
    sumsq 'dummy objective'
    fit(id)
;
sumsq.. sse =n= 0;
fit(i).. y(i) =e= sum(j, x(i,j)*coeff(j));
model ols /sumsq,fit/;
option lp=ls;
solve ols using lp minimizing sse;
display "OLS solution",coeff.l;

*-----
* Tobit model
*-----

variables
    loglike
    sigma
;
equations
    objective
;
objective.. loglike ===
    sum(i$(y(i)>0), -0.5*sqr[y(i)-sum(j,x(i,j)*coeff(j))]/sqr(sigma) ) +
    sum(i$(y(i)=0), log(1-errorf(sum(j,x(i,j)*coeff(j))/sigma))) -
    sum(i$(y(i)>0), log(sqrt(2*pi)));
*
* initial value
*
sigma.l = 1;
model tobit /objective/;
solve tobit using nlp maximizing loglike;
display "Tobit model",coeff.l;

```

The reported solution is:

```

---- 680 OLS solution
---- 680 VARIABLE coeff.L
const 5.608,    Z2   -0.050,    Z3    0.162,    Z5   -0.476,    Z7    0.106,    Z8   -0.712
---- 711 Tobit model
---- 711 VARIABLE coeff.L
const 8.174,    Z2   -0.179,    Z3    0.554,    Z5   -1.686,    Z7    0.326,    Z8   -2.285

```

The same model estimated with Gretl gives:

```

gretl version 1.3.0
Current session: 2005/01/06 18:21
# Fair's extra-marital affairs data
? open greene22_2.gdt

Read datafile /usr/share/gretl/data/greene/greene22_2.gdt
periodicity: 1, maxobs: 601,
observations range: 1-601

```

```

Listing 10 variables:
 0) const      1) Y          2) Z1          3) Z2          4) Z3
 5) Z4      6) Z5          7) Z6          8) Z7          9) Z8

# initial OLS
? ols Y 0 Z2 Z3 Z5 Z7 Z8

Model 1: OLS estimates using the 601 observations 1-601
Dependent variable: Y

      VARIABLE      COEFFICIENT      STDEROR      T STAT      2Prob(t > |T|)

 0)  const      5.60816      0.796599      7.040      < 0.00001 *** 
 3)  Z2      -0.0503473      0.0221058     -2.278      0.023107 ** 
 4)  Z3      0.161852      0.0368969      4.387      0.000014 *** 
 6)  Z5      -0.476324      0.111308     -4.279      0.000022 *** 
 8)  Z7      0.106006      0.0711007      1.491      0.136510 
 9)  Z8      -0.712242      0.118289     -6.021      < 0.00001 *** 

Mean of dependent variable = 1.45591
Standard deviation of dep. var. = 3.29876
Sum of squared residuals = 5671.09
Standard error of residuals = 3.08727
Unadjusted R-squared = 0.13141
Adjusted R-squared = 0.124111
F-statistic (5, 595) = 18.0037 (p-value < 0.00001)

MODEL SELECTION STATISTICS

SGMASQ      9.53125      AIC      9.62640      FPE      9.62640
HQ      9.79236      SCHWARZ      10.0585      SHIBATA      9.62450
GCV      9.62736      RICE      9.62834

Excluding the constant, p-value was highest for variable 8 (Z7)

# Tobit version
? tobit Y 0 Z2 Z3 Z5 Z7 Z8
Convergence achieved after 100 iterations

Model 2: Tobit estimates using the 601 observations 1-601
Dependent variable: Y

      VARIABLE      COEFFICIENT      STDEROR      T STAT      2Prob(t > |T|)

 0)  const      8.17411      2.60908      3.133      0.001816 *** 
 3)  Z2      -0.179330      0.0757210     -2.368      0.018189 ** 
 4)  Z3      0.554137      0.140708      3.938      0.000092 *** 
 6)  Z5      -1.68621      0.413967     -4.073      0.000053 *** 
 8)  Z7      0.326052      0.264723      1.232      0.218558 
 9)  Z8      -2.28496      0.443770     -5.149      < 0.00001 *** 

Mean of dependent variable = 1.45591
Standard deviation of dep. var. = 3.29876
Censored observations: 451 (75.0%)
sigma = 8.24703
Log-likelihood = -705.576

Test for normality of residual -
Null hypothesis: error is normally distributed
Test statistic: Chi-square(2) = 14.3781
with p-value = 0.000754799

```

Some authors suggest a transformation that guarantee global optima [40] while other evidence suggests this is not needed [23].

4.4. GARCH: Constrained maximum likelihood estimation for Generalized Autoregressive Conditional Heteroscedasticity models. GARCH or Generalized Autoregressive Conditional Heteroskedasticity[10] models have become

a popular tool in describing and forecasting financial time series with fluctuating volatility[11]. The GARCH(p, q) model can be stated as:

$$(40) \quad Y_t = F(\beta, X_t) + \epsilon_t$$

$$(41) \quad \epsilon_t = \sigma_t z_t$$

$$(42) \quad \sigma_t = \sqrt{h_t}$$

$$(43) \quad h_t = \alpha + \sum_{i=1}^p \delta_i h_{t-i} + \sum_{j=1}^q \gamma_j \epsilon_{t-j}^2$$

where Y_t are the observations and $F(\beta, X_t) = \mu$ is a function of independent variables (which we assume to be simply the constant function in our example). The residuals ϵ_t form a stochastic process described by the latter three equations, where z_t follows in our case a Gaussian distribution.

We want to estimate $\theta = (\mu, \alpha, \delta_i, \gamma_j)$. The standard way of doing is to maximize the likelihood function

$$(44) \quad L_T(\theta) = \sum_t \ell_t(\theta)$$

with

$$(45) \quad \ell_t(\theta) = \ln f_z(\epsilon_t | \sigma_t)$$

As z_t is Gaussian, we have

$$(46) \quad f_z(\epsilon_t | \sigma_t) = \frac{1}{\sqrt{2\pi}} \frac{1}{\sigma} \exp \left\{ -\frac{1}{2} \left(\frac{\epsilon_t}{\sigma} \right)^2 \right\}$$

Stationarity of the process requires the following conditions on the parameters: $\alpha > 0$, $\delta_i \geq 0$, $\gamma_j \geq 0$ and $\sum_i \delta_i + \sum_j \gamma_j \leq 1$. Thus the following optimization model can be formulated:

GARCHML	$\underset{\mu, \alpha, \delta_i, \gamma_j}{\text{maximize}} \quad \sum_t -\frac{1}{2} \ln(2\pi) - \frac{1}{2} \ln(h_t) - \frac{1}{2} \frac{\epsilon_t^2}{h_t}$ $\text{subject to} \quad \sum_i \delta_i + \sum_j \gamma_j \leq 1$ $\alpha > 0, \delta_i \geq 0, \gamma_j \geq 0$
----------------	---

A slight rewrite gives as objective[35]

$$(47) \quad \underset{\mu, \alpha, \delta_i, \gamma_j}{\text{minimize}} \quad \sum_t \ln(h_t) + \frac{\epsilon_t^2}{h_t}$$

Changing the stationarity condition into

$$(48) \quad \sum_i \delta_i + \sum_j \gamma_j = 1$$

results in the IGARCH model (Integrated GARCH).

For a description of the use of SQP (Sequential Quadratic Programming) to estimate these type of models see [44].

The model below is using data from [22]. We use a standard formulation for the returns:

$$(49) \quad Y_t = 100 \ln \frac{S_t}{S_{t-1}}$$

where S_t is the original time series. The model is fairly inefficient: it contains many more variables and constraints than is strictly needed if we could form e_t and h_t inside the objective function evaluation.




FIGURE 7. Time series data Y_t

4.4.1. Model garch.gms.¹⁵

```
$title GARCH -- Generalized Autoregressive Conditional Heteroskedasticity
$ontext

Example of restricted maximum likelihood estimation.

Erwin Kalvelagen, 2001

References:
  Jon Kierkegaard, Estimation of Nonlinear Stochastic Processes,
  MSc Thesis, Technical University of Denmark, report
  IMM-EKS-2000-16, April 2000.

  Bollerslev, T. (1986), "Generalized Autoregressive Conditional
  Heteroskedasticity", Journal of Econometrics 31, 307-327.

  Bollerslev, T., Chou, R.Y. and Kroner, K.F. (1992), "ARCH
  Modeling in Finance: A Review of the Theory and Empirical
  Evidence," Journal of Econometrics 52, 5-59.

  Data set from: Stephen F. Gray, GARCH Code,
  http://www.duke.edu/~sg12/software/garch/garch.htm

$offtext

----- order of the GARCH(p,q) model -----
```

¹⁵<http://amsterdamoptimization.com/models/statistics/garch.gms>


```

h(t)      'sigma(t)=sqrt(h(t))'
e(t)      'residuals'
alpha_0   'GARCH parameter (constant)'
alpha(q)  'GARCH parameters (q)'
beta(p)  'GARCH parameters (p)'
constant 'information structure: y = constant + e'
;

equations
  loglike    'transformed log likelihood function'
  def_h(t)   'h(t) structure'
  stat_garch 'stationarity condition (GARCH model)'
  stat_igarch 'stationarity condition (IGARCH model)'
  series(t)  'defines the time series'
;
loglike..  L =e= sum(lag(t), log(h(t)) + sqr(e(t))/h(t));
def_h(lag(t)).. h(t) =e= alpha_0 + sum(q, alpha(q)*sqr(e(t-ord(q)))) +
                     + sum(p, beta(p)*h(t-ord(p)));
stat_garch.. sum(q, alpha(q)) + sum(p, beta(p)) =l= 1;
stat_igarch.. sum(q, alpha(q)) + sum(p, beta(p)) =e= 1;
series(lag(t)).. y(t) =e= constant + e(t);

*
* lower bounds
*
alpha_0.lo = 0;
alpha.lo(q) = 0;
beta.lo(p) = 0;
h.lo(t) = 0.01;

*
* upper bounds
*
alpha_0.up = 100;
alpha.up(q) = 100;
beta.up(p) = 100;
h.up(t) = 100;

*
* initial values
*
alpha_0.l = .1;
alpha.l(q) = .1;
beta.l(p) = .1;
h.l(t) = .1;
e.l(t) = .1;

model garch /loglike, def_h, stat_garch, series/;
model igarch /loglike, def_h, stat_igarch, series/;

solve garch using nlp minimizing L;
solve igarch using nlp minimizing L;

```

5. M REGRESSION

M-estimation, introduced by [27], is a form of “robust” regression that is a mix of least squares estimation and LAD estimation (see section 2). It uses a least squares criterion when the residuals are small, but switches to a LAD measure for the large residuals. The Huber M-estimation problem is stated as:

M-ESTIMATION	$\begin{aligned} & \text{minimize}_{\beta} \sum_i \rho(\epsilon_i) \\ & \text{subject to } y_i = \sum_j x_{i,j} \beta_j + \epsilon_i \end{aligned}$
---------------------	---

where $\rho(\cdot)$ is defined as

$$(50) \quad \rho(\epsilon_i) = \begin{cases} \epsilon_i^2 & \text{if } |\epsilon_i| \leq k \\ 2k|\epsilon_i| - k^2 & \text{otherwise} \end{cases}$$

for some $k > 0$.





FIGURE 8. Graph of the function $\rho(x)$, $k = 1.5$.

Although this model is perfectly smooth and has continuous derivatives it is not easily formulated in GAMS. The objective function would require an if-then-else construct which is not part of the GAMS language. A direct formulation therefore would require the use of discrete variables, which makes the model an MINLP model.

In [18] and [52] a linear complementarity model is conceived as follows. Write the problem as

$$(51) \quad \min \sum_{i=1}^{\ell} \rho([Ax - b]_i)$$

$$\rho(t) = \begin{cases} \frac{1}{2}t^2 & |t| \leq \gamma, \\ \gamma|t| - \frac{1}{2}\gamma^2 & |t| > \gamma \end{cases}$$

FIGURE 9. Graph of the functions that form $\rho(x)$, $k = 1.5$.

Note that the parameters to be estimated are x here. The data is stored in A and b . If we write this as:

$$(52) \quad \min \sum_{i=1}^{\ell} \rho(v_i) \\ v = Ax - b$$

then we can form the Lagrangean:

$$(53) \quad \mathcal{L}(x, v, \lambda) = \sum_{i=1}^{\ell} \rho(v_i) - \lambda^T(v - Ax + b)$$

If we set the derivates to zero, we get:

$$(54) \quad \begin{aligned} \frac{\partial \mathcal{L}}{\partial x} &= 0 \Rightarrow A^T \lambda = 0 \\ \frac{\partial \mathcal{L}}{\partial v} &= 0 \Rightarrow \frac{\partial \rho(v_i)}{\partial v_i} - \lambda_i = 0 \\ \frac{\partial \mathcal{L}}{\partial \lambda} &= 0 \Rightarrow v = Ax - b \end{aligned}$$

The partial derivatives of $\rho(v_i)$ are as follows:

$$(55) \quad \frac{\partial \rho(v_i)}{\partial v_i} = \begin{cases} v_i & |v_i| \leq \gamma \\ \gamma \operatorname{sign}(v_i) & |v_i| > \gamma \end{cases}$$

where $\text{sign}(x)$ is defined as:

$$(56) \quad \text{sign}(x) = \begin{cases} +1 & x \geq 0 \\ -1 & x < 0 \end{cases}$$

This results in:

$$(57) \quad \frac{\partial \mathcal{L}}{\partial v} = 0 \Rightarrow \begin{cases} \lambda_i - [Ax - b]_i = 0 & |v_i| \leq \gamma \\ \lambda_i - \gamma = 0 & v_i \geq \gamma \\ \lambda_i + \gamma = 0 & -v_i \geq \gamma \end{cases}$$

The whole system can now be written as a mixed linear complementarity problem:

$$(58) \quad \begin{aligned} \lambda - Ax + b + s^+ - s^- &= 0 \\ A^T \lambda &= 0 \\ \lambda + \gamma e \geq 0 \perp s^+ \geq 0 \\ -\lambda + \gamma e \geq 0 \perp s^- \geq 0 \end{aligned}$$

where \perp is used to indicate complementarity between an equation and a variable.

The complete LCP model is listed below. In the model we use the data from section 1.1.

5.0.2. Model huberlcp.gms.¹⁶

```
$ontext
Huber M regression by LCP
Erwin Kalvelagen, november 2001

$offtext

set i 'number of cases' /i1*i40/;
set j 'coefficient to estimate' /'constant','coeff1'?;

$include expdata.inc

parameter b(i);
b(i) = data(i,'expenditure');

parameter A(i,j);
A(i,'constant') = 1;
A(i,'coeff1') = data(i,'income');

scalar gamma /1.5/;

variables
  w(i)   'lagrange multipliers'
  x(j)   'parameters to estimate'
  lambda1(i) 'slack'
  lambda2(i) 'slack'
;
positive variables lambda1, lambda2;

equations
  e1(i)
  e2(j)
  compl1(i)
  compl2(i)
;
e1(i).. w(i) - sum(j, A(i,j)*x(j)) + b(i) + lambda2(i) - lambda1(i) =e= 0;
```

¹⁶<http://amsterdamoptimization.com/models/statistics/huberlcp.gms>

```

| e2(j).. sum(i, A(i,j)*w(i)) =e= 0;
| compl1(i).. w(i) + gamma =g= 0;
| compl2(i).. -w(i) + gamma =g= 0;
|
model m /e1,e2,compl1.lambda1,compl2.lambda2/;
solve m using mcp;

```

It is sometimes suggested to use $k = 1.5\hat{\sigma}$, with

$$(59) \quad \hat{\sigma} = 1.483 \text{ MAD}$$

where MAD is the median of the absolute deviations $|\epsilon_i|$.

For more information on M estimation and robust statistics see for instance [28].

6. MAXIMUM ENTROPY METHODS

The availability of data is often a problem in practical applied general equilibrium modeling. The lack of data can prevent the use of standard regression techniques to estimate model parameters. A recent technique called *Maximum Entropy*[20] has become a popular device in this field to get estimates[43, 42, 2].

Let X be a random variable with a discrete distribution $P(X = x_k) = \pi_k$ for $k = 1, \dots, N$, where $\sum_k \pi_k = 1$. The entropy-information measure introduced by [45] is defined by

$$(60) \quad S(\pi) = - \sum_{k=1}^N \pi_k \ln(\pi_k)$$

Following [12] we consider again the general linear model (GLM):

$$(61) \quad y = X\beta + e$$

Suppose we have prior information in the form of bounds on the parameters β_k and on the error terms. Let

$$(62) \quad z_{k,1} \leq \beta_k \leq z_{k,2}$$

then we can write the linear combination:

$$(63) \quad \beta_k = p_k z_{k,1} + (1 - p_k) z_{k,2} = (z_{k,1} \ z_{k,2}) \begin{pmatrix} p_k \\ 1 - p_k \end{pmatrix}$$

with $p_k \in [0, 1]$. Extending this to M “support points” for each parameter β_k , we get the following linear system:

$$(64) \quad \begin{aligned} \beta_k &= \sum_{j=1}^M z_{k,j} p_{k,j} \\ \sum_{j=1}^M p_{k,j} &= 1 \\ p_{k,j} &> 0 \end{aligned}$$

Similarly we can write a set of J support points for each error term e_i :

$$(65) \quad \begin{aligned} e_i &= \sum_{j=1}^J v_{i,j} w_{i,j} \\ \sum_{j=1}^J w_{i,j} &= 1 \\ w_{i,j} &> 0 \end{aligned}$$

Using matrix notation we can now write:

$$(66) \quad y = XZp + Vw$$

The Maximum Entropy Estimation problem can now be stated as:

$$(67) \quad \begin{aligned} \max H(p, w) &= -p^T \ln(p) - w^T \ln(w) \\ y &= XZp + Vw \\ (I_K \otimes i_M^T)p &= i_K \\ (I_N \otimes i_J^T)w &= i_N \\ p, w &> 0 \end{aligned}$$

where \otimes is the Kronecker product and i_N is an N vector of ones.

The prior information is the model below is as follows. We assume a parameter support for parameter $const$ of $z_{const}^T \{-50, -25, 0, 25, 50\}$. All other parameters have a support of $z_k^T = \{-20, -10, 0, 10, 20\}$. I.e. we slightly wider bounds for the constant term coefficient. The error support is roughly $\pm 3\sigma$ with σ being the sample standard deviation. I.e. $v_i^T = \{-10, -5, 0, 5, 10\}$.

6.0.3. Model gme.gms.¹⁷

```
$ontext
Generalized Maximum Entropy
Erwin Kalvelagen, march 2003

References:
Maximum entropy estimation in economic models with linear
inequality restrictions, Randall C. Campbell, R. Carter Hill
Department of Economics, Louisiana State University,
Baton Rouge, LA 70803,USA

$offtext

set i 'cases' /case1*case116/;

set k 'parameters' /const, famsize, unemp, highschl, college, medinc, d90/;

table data(i,*)
      pov   famsize   unemp   highschl   college   medinc   d90
case1    18.1     3.15    10.8     53.7     22.3    22.863     0
case2     8.7     3.20     6.9     53.7     32.4    17.240     0
case3     7.5     2.87     7.2     64.2     12.6    18.065     0
case4     9.5     2.93    10.5     54.7     16.9    16.301     0
case5     7.5     2.88     9.6     62.5     13.8    17.909     0
case6     8.8     3.18     8.2     52.3     12.3    17.842     0
case7     6.1     3.16     5.8     56.2     25.5    26.513     0
```

¹⁷<http://amsterdamoptimization.com/models/statistics/gme.gms>


```

| case80    10.7   2.42    6.3   61.0   16.8   29.468   1
| case81    11.0   2.57   10.9   60.9   17.8   31.276   1
| case82    15.4   3.17   14.6   51.1   12.0   28.269   1
| case83    11.6   2.49   12.4   61.0   11.2   27.407   1
| case84     6.7   2.48   12.5   65.9   21.9   35.932   1
| case85     8.5   2.96   10.9   51.4   21.5   36.223   1
| case86     4.6   2.54    5.9   58.4   22.3   42.789   1
| case87     5.8   2.51    7.0   64.2   22.1   36.942   1
| case88     5.2   2.87    4.8   53.4   27.8   51.167   1
| case89     5.3   2.66    6.8   62.4   22.7   42.805   1
| case90     9.8   2.41   12.0   67.6   15.1   29.967   1
| case91     8.4   2.85   10.7   59.5   14.6   37.694   1
| case92     9.8   2.58    6.3   59.2   23.0   37.841   1
| case93     7.3   3.15   17.2   54.0   14.4   39.637   1
| case94    10.3   2.97    8.0   60.5   14.9   36.977   1
| case95     8.1   2.69    6.1   56.6   25.3   39.798   1
| case96     9.7   2.29    5.6   43.0   35.0   40.561   1
| case97    12.0   2.94   12.0   55.4   13.2   34.701   1
| case98     6.8   2.53    5.8   60.4   22.9   37.086   1
| case99     4.3   2.64    4.2   52.8   31.3   53.430   1
| case100    7.4   2.73    6.0   53.4   26.6   41.289   1
| case101    5.0   2.81    5.5   49.4   32.6   53.670   1
| case102    6.2   2.66    8.0   52.2   29.7   43.130   1
| case103    11.0   2.58   10.3   64.7   13.7   30.332   1
| case104    5.7   2.45   10.5   59.6   15.9   29.911   1
| case105    11.6   2.48   12.5   63.2   14.2   26.073   1
| case106    6.0   2.88    7.0   64.0   18.7   42.392   1
| case107    5.2   2.55    5.7   59.9   24.5   41.961   1
| case108    11.4   2.91   14.3   55.4   13.0   32.923   1
| case109    12.2   2.75   17.6   56.9   15.4   31.842   1
| case110    12.6   2.60   12.4   62.0   10.2   25.946   1
| case111    15.1   2.49   14.5   61.3   12.9   25.009   1
| case112    18.0   3.12   17.1   48.4   11.8   26.697   1
| case113    6.9   2.46    8.3   65.3   14.7   31.464   1
| case114    5.0   3.02    7.0   56.4   23.0   50.091   1
| case115    9.8   2.63    7.2   48.8   30.3   36.866   1
| case116   16.0   2.85   14.1   59.0    9.5   24.364   1
;

parameters X(i,k) 'exogenous matrix';
X(i,k) = data(i,k);
X(i,'const') = 1;
display X;

parameter y(i) 'endogenous variable';
y(i) = data(i,'pov');
display y;

*-----
* OLS model
*-----

variables b(k);
variables sse, e(i);
equations sumsq,linear(i);

linear(i)..  y(i) =e= sum(k, X(i,k)*b(k)) + e(i);
sumsq..      sse =e= sum(i, sqr(e(i)));

model ols /linear,sumsq;
solve ols minimizing sse using nlp;

display b,l,sse,l;

*-----
* GME model
*-----


set j /j1*j5/;
table z(k,j) 'parameter support for GME model'
      j1    j2    j3    j4    j5
const      -50    -25     0    25    50

```

```

famsize    -20   -10     0    10    20
unemp      -20   -10     0    10    20
highschl   -20   -10     0    10    20
college    -20   -10     0    10    20
medinc     -20   -10     0    10    20
d90        -20   -10     0    10    20
;

parameter errsupport(j) 'error support' /
j1 -10
j2 -5
j3 0
j4 5
j5 10
/;

parameter v(i,j);
v(i,j) = errsupport(j);

variables p(k,j), w(i,j);
p.lo(k,j) = 0.0001;
w.lo(i,j) = 0.0001;

variable entrpy;
equations
parmsupp(k) 'parameter support',
errsupp(i) 'error support'
normp(k)   'normalize p'
normw(i)   'normalize w'
obj        'maximize entropy'
;

obj..      entrpy == -sum((k,j), p(k,j)*log(p(k,j)))-sum((i,j), w(i,j)*log(w(i,j)));
parmsupp(k).. b(k) == sum(j, z(k,j)*p(k,j));
errsupp(i).. e(i) == sum(j, v(i,j)*w(i,j));
normp(k)..   sum(j, p(k,j)) == 1;
normw(i)..   sum(j, w(i,j)) == 1;

model gme /obj,parmsupp,errsupp,linear,normp,normw/;
solve gme maximizing entrpy using nlp;

display b.l,entrpy.l;

*-----*
* IRLS model (inequality restricted least squares)
*-----*

*
* add sign restriction on college coefficient as it has the wrong
* sign in the OLS estimates.
*
b.up('college') = 0;

solve ols minimizing sse using nlp;
display b.l,sse.l;

* repair
b.up('college') = INF;

*-----*
* RGME model
*-----*

*
* introduce sign restrictions on the parameters
* by providing appropriate parameter support
*
table z2(k,j) 'parameter support for GME model'
j1    j2    j3    j4    j5
const    -50   -25     0    25    50

```

solver	seconds
MINOS 5.51	1.4
SNOPT 6.2	1.4
CONOPT 3	2.6
PATHNLP	0.1

TABLE 3. Time for first GME model

```

famsize      0      5     10    15    20
unemp        0      5     10    15    20
highschl   -20    -15    -10    -5     0
college     -20    -15    -10    -5     0
medinc     -20    -15    -10    -5     0
d90        -20    -10      0    10    20
;
z(k,j) = z2(k,j);

solve gme maximizing entryp using nlp;
display b.1,entryp.1;

```

The last two models deal with sign restrictions on the parameters. For the GME model we can do this indirectly by providing a support that is not around zero but rather one sided. The data for this model is from [41].

The references [43, 42] includes complete GAMS code showing how these techniques can be used to estimate a SAM (Social Accounting Matrix) opposed to the more traditional RAS method. In [39] it is emphasized that RAS is actually a form of Entropy Optimization. A large scale entropy estimation application using GAMS and PATH is documented in [21]. As the NLP models tend to have many superbasic variables for large some instances, reformulating the optimization model into a complementarity problem can lead to a large performance gain. With the PATHNLP solver this process is completely automated. Indeed in the above example the entropy models solve very fast by using the PATHNLP solver compared to the traditional NLP solvers MINOS, SNOPT, and CONOPT as shown in table 6.0.3.

6.1. Confidence intervals for max entropy models. Maximum entropy models do not provide standard errors or confidence intervals. A bootstrap (resampling) approach can be used to calculate those. A complete example is given in [33].

7. CLASSIFICATION

A problem related used in Machine Learning and Data Mining is the classification problem where we try to find a simple rule to separate data points[37, 16]. I.e. given two sets of data points A and B we would like to find a discriminant function f such that $f(x) < 0$ for $x \in A$ and $f(x) > 0$ for $x \in B$.

The simplest case is the linear classification problem, as depicted in figure 7. Mathematically we are seeking to find a plane $x^T w = \gamma$ with the property that all data points $x \in A$ have $x^T w > \gamma$ and all data points in B have $x^T w < \gamma$ or

$$(68) \quad \begin{aligned} Aw &\geq e\gamma + e \\ Bw &\leq e\gamma - e \end{aligned}$$




FIGURE 10. Linear separation

where e is a vector of ones. The following linear program[38]:

$$(69) \quad \begin{aligned} & \min_{w, \gamma, y, z} \frac{e^T y}{n} + \frac{e^T z}{m} \\ & Aw + y \geq e\gamma + e \\ & Bw - z \leq e\gamma - e \\ & y, z \geq 0 \end{aligned}$$

will find a separating hyperplane if it exists. In this case $y = 0$, $z = 0$. If the points can not be separated by a hyperplane, it will find a plane that minimizes the average sum of the violations. Another advantage of this formulation is that $w = 0$ is naturally eliminated. Other formulations are discussed in [19, 26].

7.0.1. Model *classify.gms*.¹⁸

```
$ontext
classification through linear programming

Reference:
  O. L. Mangasarian, W. N. Street, and W. W. Wolberg,
  "Breast Cancer Diagnosis and Prognosis Via Linear Programming,"
  Operations Research, 43 (1995), 570-577

Data from http://cgmm.cs.mcgill.ca/~beezer/cs644/example.html

$offtext

set
d '2d dataset' /x,y/
i 'cases for points A' /i1*i13/
j 'cases for points B' /j1*j17/
;

table pa(i,d)
  x      y
i1    3      1
```

¹⁸<http://amsterdamoptimization.com/models/statistics/classify.gms>

```

i2 2 2
i3 4 3
i4 3 4
i5 0 5
i6 1 5
i7 5 5
i8 1 6
i9 2 7
i10 3 7
i11 1 8
i12 0 9
i13 2 10
;

table pb(j,d)
  x y
j1 9 0
j2 10 0
j3 10 1
j4 10 3
j5 8 4
j6 12 4
j7 9 5
j8 11 5
j9 6 6
j10 8 6
j11 10 6
j12 8 7
j13 9 7
j14 8 8
j15 5 8
j16 5 9
j17 7 9
;

scalar n,m;
n = card(i);
m = card(j);

variables obj,gamma,w(d);
positive variables y(i),z(j);
equations objdef,eqa(i),eqb(j);

objdef.. obj =e= sum(i,y(i))/n + sum(j,z(j))/m;
eqa(i).. sum(d,pa(i,d)*w(d)) + y(i) =g= gamma+1;
eqb(j).. sum(d,pb(j,d)*w(d)) - z(j) =l= gamma-1;

model classify /objdef,eqa,eqb/;
solve classify minimizing obj using lp;

display gamma.l,w.l;

```

When we replace the data by:

```

set
d '2d dataset' /x,y/
i 'cases for points A' /i1*i16/
j 'cases for points B' /j1*j20/
;

table pa(i,d)
  x y
i1 3 1
i2 2 2
i3 4 3
i4 3 4
i5 0 5
i6 1 5
i7 5 5
i8 1 6
i9 2 7

```




FIGURE 11. Results of classify.gms.

```
i10 3 7
i11 1 8
i12 0 9
i13 2 10
i14 6 7
i15 7 5
i16 6 4
;
table pb(j,d)
  x y
j1 9 0
j2 10 0
j3 10 1
j4 10 3
j5 8 4
j6 12 4
j7 9 5
j8 11 5
j9 6 6
j10 8 6
j11 10 6
j12 8 7
j13 9 7
j14 8 8
j15 5 8
j16 5 9
j17 7 9
j18 5 3
j19 3 9
j20 8 3
;
```

the points can not be separated linearly. The results are depicted in figure 12.




FIGURE 12. Data can not be separated linearly

REFERENCES

1. R. D. Armstrong and M. T. Kung, *An algorithm to select the best subset for a least absolute value regression problem*, Optimization in Statistics (S. H. Zanakis and J. S. Rustagi, eds.), Studies in the Management Sciences, vol. 19, North-Holland Publishing Company, 1982, pp. 67–80.
2. Channing Arndt, Sherman Robinson, and Finn Tarp, *Parameter Estimation for a Computable General Equilibrium Model: A Maximum Entropy Approach*, <http://www.ifpri.org/divs/tmd/dp/papers/tmdp40.pdf>, February 1999.
3. K. Arrow, H. Chenery, B. Minhas, and R. Solow, *Capital-Labor Substitution and Economic Efficiency*, Review of Economics and Statistics **43** (1961), 225–250.
4. Jushan Bai, *Least Absolute Deviation Estimation of a Shift*, Econometric Theory **11** (1995), no. 3, 403–436.
5. Jushan Bai and Pierre Perron, *Testing for and Estimation of Multiple Structural Changes*, Econometrica (1998), 47–78.
6. ———, *Computation and Analysis of Multiple Structural Change Models*, Journal of Applied Econometrics **18** (2003), no. 1, 1–22.
7. Herman J. Bierens, *The Tobit model*, September 2004.
8. D. Birkes and Y. Dodge, *Alternative Methods of Regression*, Wiley, 1993.
9. P. Bloomfield and W. L. Steiger, *Least Absolute Deviations. Theory, Applications and Algorithms*, Birkhäuser, 1983.
10. T. Bollerslev, *Generalized Autoregressive Conditional Heteroskedasticity*, Journal of Econometrics **31** (1986), 307–327.
11. T. Bollerslev, R. Y. Chou, and K. F. Kroner, *ARCH Modeling in Finance*, Journal of Econometrics **52** (1992), 5–59.
12. Randall C. Campbell and R. Carter Hill, *Maximum Entropy Estimation in Economic Models with Linear Inequality Restrictions*, Tech. Report 2001-11, Louisiana State University, Department of Economics, 2001.

13. V. Chvatal, *Linear Programming*, Freeman, 1983.
14. D. R. Cox and E. J. Snell, *Applied Statistics: Principles and Examples*, Chapman and Hall, London, 1981.
15. Terry Dielman and Roger Pfaffenberger, *LAV (least absolute value) estimation in linear regression: a review*, Optimization in Statistics (S. H. Zanakis and J. S. Rustagi, eds.), Studies in the Management Sciences, vol. 19, North-Holland Publishing Company, 1982, pp. 31–52.
16. R. O. Duda, P. E. Hart, and David G. Stork, *Pattern Classification*, Wiley, 2000.
17. M. Fuss, D. McFadden, and Y. Mundlak, *A survey of functional forms in the economic analysis of production*, The Theory of Production (M. Fuss and D. McFadden, eds.), Production Economics: A Dual Approach to Theory and Applications, vol. I, North-Holland, 1978.
18. Michael Gertz and Stephen Wright, *Object-oriented Software for Quadratic Programming*, Tech. Report ANL/MCS-P891-1000, Argonne National Laboratory, Mathematics and Computer Science Division, 2001.
19. F. Glover, *Improved Linear Programming Models for Discriminant Analysis*, Decision Sciences **21** (1990), 771–785.
20. Amos Golan, George G. Judge, and Douglas Miller, *Maximum Entropy Econometrics: Robust Estimation with Limited Data*, Wiley, 1996.
21. Amos Golan, Jeffrey M. Perloff, and Edward Zhihua Shen, *Estimating a Demand System with Nonnegativity Constraints: Mexican Meat Demand*, Tech. report, University of California, Berkeley, July 2000.
22. S. F. Gray, *GARCH Code*, <http://www.duke.edu/~sg12/software/garch/garch.htm>.
23. William H. Greene, *Multiple roots of the Tobit log-likelihood*, Journal of Econometrics **46** (1990), no. 3, 365–380.
24. ———, *Econometric Analysis*, 5th ed., Prentice-Hall, 2003.
25. W. E. Griffiths, R. C. Hill, and G. G. Judge, *Learning and Practicing Econometrics*, Wiley, 1993.
26. R. C. Grinold, *Mathematical Methods for Pattern Classification*, Management Science **19** (1972), 272–289.
27. P. J. Huber, *Robust Estimation of a Location Parameter*, Annals of Mathematical Statistics **35** (1964), 73–101.
28. ———, *Robust Statistics*, Wiley, 1981.
29. Jr. John E. Dennis, David M. Gay, and Roy E. Welsch, *Algorithm 573: NL2SOL – An Adaptive Nonlinear Least-Squares Algorithm*, ACM Trans. Math. Softw. **7** (1981), no. 3, 369–383.
30. Dale W. Jorgenson, *Econometrics: Econometric Modeling of Producer Behavior*, MIT Press, 2000.
31. Erwin Kalvelagen, *A Linear Regression Solver for GAMS*, <http://amsterdamoptimization.com/pdf/regression.pdf>, 2004.
32. ———, *A Non-Linear Regression Solver for GAMS*, <http://amsterdamoptimization.com/pdf/nlregression.pdf>, 2007.
33. ———, *Bootstrap code for forming confidence intervals in Max Entropy estimation*, <http://amsterdamoptimization.com/pmwiki/pmwiki.php?n=Download.Bootstrap>, 2007.
34. ———, *Critical values for the Student's t distribution*, <http://amsterdamoptimization.com/pmwiki/pmwiki.php?n=Download.Qt>, 2007.
35. J. Kierkegaard, *Estimation of Nonlinear Stochastic Processes*, M.Sc. thesis IMM-EKS-2000-16, Technical University of Denmark, April 2000.
36. H. van Maaren and T. Terlaky, *Inverse barriers and CES-functions in linear programming*, Tech. Report 95-76, Delft University of Technology, 1995.
37. O. L. Mangasarian, *Linear and Nonlinear Separation of Patterns by Linear Programming*, Operations Research **13** (1965), 444–452.
38. O. L. Mangasarian, W. N. Street, and W. W. Wolberg, *Breast Cancer Diagnosis and Prognosis Via Linear Programming*, Operations Research (1995), no. 43, 570–577.
39. Robert McDougall, *Entropy Theory and RAS are Friends*, GTAP Working Paper 300, Center for Global Trade Analysis, Department of Agricultural Economics, Purdue University, May 1999.
40. R. Olsen, *A Note on the Uniqueness of the Maximum Likelihood Estimator in the Tobit Model*, Econometrica **46** (1978), 1211–1215.

41. R. Ramanathan, *Introductory econometrics with applications*, (Harcourt College Publishers, 2002).
42. Sherman Robinson, Andrea Cattaneo, and Moataz El-Said, *Estimating a Social Accounting Matrix Using Cross Entropy Methods*, <http://www.ifpri.org/divs/tmd/dp/papers/tmdp33.pdf>, October 1998.
43. Sherman Robinson and Moataz El-Said, *Estimating a Social Accounting Matrix Using Entropy Difference Methods*, <http://www.ifpri.org/divs/tmd/dp/papers/tmdp21.pdf>, September 1997.
44. R. Schoenberg, *Constrained Maximum Likelihood*, Tech. report, Aptech and The University of Washington, November 1996.
45. C. E. Shannon, *A Mathematical Theory of Communication*, Bell System Technical Journal **27** (1948), 379–423.
46. S. M. Stigler, *The History of Statistics. The Measurement of Uncertainty before 1900*, The Belknap Press of Harvard Press, 1986.
47. H. Theil, *Principles of Econometrics*, Wiley, 1971.
48. Luke Tierney, *XLISP-STAT, A Statistical Environment Based on the XLISP Language (Version 2.0)*, Tech. Report 528, University of Minnesota, School of Statistics, July 1989.
49. J. Tobin, *Estimation of relationships for limited dependent variables*, *Econometrica* **26** (1958), 24–36.
50. Hal R. Varian, *Intermediate Microeconomics. A modern approach*, fifth ed., W. W. Norton & Co., 1999.
51. Diana Whistler, Kenneth J. White, S. Donna Wong, and David Bates, *SHAZAM Version 9 User's Reference Manual*, Northwest Econometrics, 2001.
52. Stephen J. Wright, *On reduced convex QP formulations of monotone LCPs*, *Mathematical Programming* **90** (2001), 459–473.

AMSTERDAM OPTIMIZATION MODELING GROUP, WASHINGTON DC

E-mail address: erwin@amsterdamoptimization.com